PROBLEMY INŻYNIERII ROLNICZEJ 2016 (I–III): z. 1 (91)
PROBLEMS OF AGRICULTURAL ENGINEERING s. 27–40

Wersja pdf: www.itp.edu.pl/wydawnictwo/pir/ ISSN 1231-0093

Wpłynęło 31.12.2015 r.
Zrecenzowano 01.02.2016 r.
Zaakceptowano 02.02.2016 r.

A – koncepcja
B – zestawienie danych
C – analizy statystyczne
D – interpretacja wyników
E – przygotowanie maszynopisu
F – przegląd literatury

Study of kinematics of an agricultural machine by the Maple program

Jan ČERVINKA[1] ACD, Stanislav BARTOŇ[1] BCEF, Edmund KAMIŃSKI[2] DEF, Bronisław PUCZEL[3] BEF

[1] *Mendel University of Agriculture and Forestry, Brno Czech Republic*
[2] *Institute of Technology and Life Sciences in Falenty*
[3] *Experimental Station for Cultivar Testing in Krzyżewo*

Do cytowania For citation: Červinka J., Bartoň S., Kamiński E., Puczel B. 2016. Study of kinematics of an agricultural machine by the Maple program. Problemy Inżynierii Rolniczej. Z. 1 (91) s. 27–40.

Summary

This article presents principles of application of the program Maple 7 to analyze the kinematics of a complicated machine. Trajectory, velocity and acceleration of the machine element have been studied. Authors carried into effect adequate computer calculations and presented their results in graphical and dynamic form. Advantages of the mathematical solution of kinematical problem using the Maple 7 program have been presented on example of pick-up drum. The method can be used for functional junction of machinery aggregates or to calculate durability of particular elements and their size. The calculations can help to reveal critical points of movement of matter into press mechanism for already manufactured machines. The right functions are possible to be verified by computer animation of initiated kinematics mechanism.

Key words: analog technique, program Maple, drum gatherer, kinematics

Introduction

Practice of recent years shows that modeling of kinematics' systems and of different objects as well as of research of their kinematics' and dynamic characteristics are continually topical. For computer animation of mechanism's kinematics the programs like: Autodesk 3ds Max – earlier 3D Studio MAX [GECOW 2012; ZAWADZKI *et al*. 2011], SAP CATIA System of Automatic Designing Computer Aided Three-dimensional Interactive Application – system of automatic designing SAP of French firm Dassault Systemes) [MALYGIN 2012], T-FLEX CAD [IGNATEV *et al*. 2015; SAPRONOV 2015], Maple are used [MITUŚ *et al*. 2007; OMBACH 2006]. When computer animation of a mechanism kinematics animation, the programs like: Autodesk 3ds Max – earlier 3D Studio MAX are applied [GECOW 2012; ZAWADZKI *et al*. 2011] are applied. Importance of computer programs in modeling kinematics' systems has been emphasized in many articles and manuals [CALKINS *et al*. 2000; CAMPION *et al*. 1996; JAKUBIAK, MUSZYŃSKI (tech. ed.) 2012; PAZDERSKI 2012; PAZDERSKI, KOZŁOWSKI 2008; YOUNG, FREEDMAN 1996]. Program Maple were applied, among other things, while research concerning bending of beams [HAŁAT 2007; ZAWADZKI *et al*. 2013], modeling of kinematics and dynamics of the hydraulic lift [DINDORF *et al*. 2000]. System of computer algebra Maple is a product of Waterloo Maple Inc. (Canada). This company elaborates computer programs for complicated mathematical calculations, data visualization and modeling since 1984. Maplesoft is offered both in versions for students and for professionalists. The newer version of Maple 7 includes first of all complete set of program packets: CurveFitting, PolinomialTools, OrthogonalSeries and other.

Trajectory of the propelled machine element

To describe its trajectory we shall use parametrical description, [$x(t)$, $y(t)$], where time t will be used as a parameter. Trajectory consists of four separate segments – two line segments and two semicircles. In this case it is best to take advantage of the piecewise defined functions. The whole problem can be solved analytically, but Maple answers will be too long, so we shall assign numerical values to the variables. We shall use font Helvetica to highlight Maple commands and slanted font to highlight the Maple answers. The SI units will be used. The origin of the coordinate system is in the center of the first teeth wheel. Signification of the used variables is shown at the Figure 1, and o = turns of the first wheel per second and pi = numerical value of π.

```
> restart; with(plots): Su:=[Cx=0.64, r=0.1925,o=2.98, Xo=2.01,Yo=-.2,X1=1.0,
  Y1=-.18, phi=35*pi/180, S=1.3, Ph[1]=0.1, Ph[2]=0.3, Ph[3]=1.085, Lh=0.4,
  pi=evalf(Pi)]: assign(Su): omega:=2*pi*o: v:=omega*r:
```

ω = angular velocity of the first wheel and v = peripheral velocity of the first wheel = sliding velocity of the chain.



Source: own elaboration. Źródło: opracowanie własne.

Fig. 1. Kinematics' scheme of a pick-up drum
Rys. 1. Schemat kinematyczny podbieracza bębnowego

At first, we shall prepare partial functions describing the trajectory of the propelled Machine element as a parametric function of time.

> L:=[Cx,r*pi,Cx,r*pi]: L:=[seq(sum(L[i],i=1..j),j=1..4)]: *L* = length of the individual trajectory segments and later cumulative length of the trajectory segments
> Tau:=map(u->u/v,L): *T* = times of the transition from the one to other trajectory segment
> Tf:=Tau[4]: *Tf* = duration of one period
> Fx[1]:=v*t: Fy[1]:=r: motion to right
> Fx[2]:=Cx+r*cos(pi/2-omega*(t-Tau[1])): Fy[2]:=r*sin(pi/2-omega*(t-Tau[1])): right semicircle
> Fx[3]:=Cx-v*(t-Tau[2]): Fy[3]:=-r: motion to left
> Fx[4]:=r*cos(3/2*pi-omega*(t-Tau[3])): Fy[4]:=r*sin(3/2*pi-omega*(t-Tau[3])): left semicircle.

Now we can build piecewise defined functions *Ξ(t)* and *H(t)*, describing trajectory. Later we shall work with them very simply – as with symbols.

> Xi:=piecewise(seq([t<=Tau[j],Fx[j]][],j=1..4));
Eta:=piecewise(seq([t<=Tau[j],Fy[j]][],j=1..4));

$$\Xi=\begin{cases} 3.6043t & t<=.1776 \\ .64+.1925\cos(-4.8954+18.7239t) & t<=.3453 \\ 1.8848-3.6044t & t<=.5229 \\ .1925\cos(-14.5033+18.7239t) & t<=.6907 \end{cases} \qquad \mathrm{H}=\begin{cases} .1925 & t<=.1776 \\ -.1925\sin(-4.8954+18.7239t) & t<=.3453 \\ -.1925 & t<=.5229 \\ -.1925\sin(-14.5033+18.7239t) & t<=.6907 \end{cases}$$

> plot([Xi,Eta,t=0..Tf],scaling=constrained): This plot command we can use as a graphical test of the correctness. The graphical output is not presented for the sake of brevity.

Trajectory of the driven machine element

Driven machine element moves along two lines. Two points give the first one. The first one is the point of intersection of both lines – here the driven machine element changes direction of its movement. The second point describes machine element's position of the first leading line. At first we have to know the equation describing this line.

> p:=y=k*x+q: p1:=solve({subs(x=Xo,y=Yo,p),subs(x=X1,y=Y1,p)},{k,q}): *p* = general equation of line, *p*1 = parameters *k* and *q* describing the first line.

The second line goes through the point of intersection and forms an angle phi with the first line. We have again to determine its parameters.

> k=subs(tan(alpha)=k,p1,expand(tan(phi-alpha))): q=solve(subs(%,x=Xo,y=Yo,p)): p2:={%,%%}: *p*1, *p*2; *p*2 = parameters *k* and *q* describing the second line

$$\{q=-.1602\ k=-0.1980\},\quad \{k=.7301,\ q=-1.6676\}$$

We shall use again piecewise function to describe position of the driven machine element as a parametric function of time. At first we have to determine when this element crosses vertex, because at this time functions will be changed. The simplest method is a numerical solution, based on the graphical output.

The length *S* of the supporting bar is always constant. As the propelled machine element moves along its trajectory [$\Xi(t)$, $H(t)$] sometimes the distance of the vertex and propelled machine element has to be equal to bars length. So we can compare the distance of the propelled machine element from the vertex with length of the bar and to plot this as a function of time.



Source: own elaboration. Źródło: opracowanie własne.

Fig. 2. Function comparing distance with bar's length
Rys. 2. Funkcja stosunku odległości bębna do długości prętów

> e:=(Xi-Xo)^2+(Eta-Yo)^2-S^2: *e* = function comparing distance with bar's length, figure 2, plot(e,t=0..Tf); zero points of this graph are times of crossing through vertex.
> tau:=[fsolve(e,t=0.2..0.3), fsolve(e,t=0.3..0.4)];

$$\tau := [.2130, .3254]$$

At time interval *τ* driven machine element moves along second line. Now we have to find analytical functions describing position of driven machine element. Generally this element moves along line, described by the parameters *k* and *q*, which will be substituted by the parameters from the variable *p*1 or *p*2. Propelled machine element has position [*X*,*Y*], which will be substituted by $\Xi(t)$ and $H(t)$ and distance of both elements = length of the bar has to be *s*, later substituted by *S*. If the driven Machine element has unknown coordinates [*x*, *y*], these coordinates has to satisfy following *eK* – condition of the constant length, the second condition is the equation *p* – driven machine element moves along line. Thus we have two equations for two variables.

> eK:=(X-x)^2+(Y-y)^2=s^2: sol:=[allvalues(solve({eK,p},{x,y}))]: we obtain two solutions so we have to select the correct one

> solf:=value(subs(X=Xi,Y=Eta,s=S,p2,t=(Tau[1]+Tau[2])*0.5,sol)): At time *t* from interval *τ* the driven machine element moves along the second line. We shall substitute *t* by the midpoint of this interval

> solf:=map(u->evalb(subs(u,x-Xo)>0),solf): and the correct solution will satisfy *x* > *Xo*

> sol:=zip((u,v)->\`if\`(u,v,NULL),solf,sol)[]: This is the correct solution

$$sol := \left\{ x = \frac{B + \sqrt{D}}{1+k^2},\ y = \frac{k\left(B + \sqrt{D}\right)}{1+k^2} + q \right\}$$

where

$$B = -kq + Yk + X$$
$$D = -2kqX + 2YkX + 2Yq - Y^2 - q^2 + s^2 - k^2X^2 + k^2s^2$$

Now we can build piecewise functions describing position of the driven machine element as a function of time.

> xi:=subs(X=X(t),Y=Y(t),s=S,piecewise(t<tau[1],subs(sol,p1,x), t<tau[2],subs(sol,p2,x),subs(sol,p1,x)));

> eta:=subs(X=X(t),Y=Y(t),s=S,piecewise(t<tau[1],subs(sol,p1,y), t<tau[2],subs(sol,p2,y),subs(sol,p1,y)));

$$\xi := \begin{cases} .9996\ X(t) - .0032 - .1979\ Y(t) + .9996\sqrt{1.6650 - .0063\ X(t) - 0.3960\ Y(t)X(t) - .320396039\ 6\ Y(t) - Y(t)^2 - 0.0004X(t)^{\ 2}} & \tau < .2130 \\ .7942 + .4762\ Y(t) + .6523\ X(t) + .6523\sqrt{2.4351X(t) + 1.4603Y(t)\ X(t) - 3.3351Y(t) - Y(t)^2 - .1899 - .5331X(t)^{\ 2}} & \tau < .3254 \\ .9996\ X(t) - .0032 - .1979\ Y(t) + .9996\sqrt{1.6650 - .0063\ X(t) - 0.3960\ Y(t)X(t) - .320396039\ 6\ Y(t) - Y(t)^2 - 0.0004X(t)^{\ 2}} & \text{otherwise} \end{cases}$$

$$\eta := \begin{cases} .0004\ Y(t) - .1601 - .0198\ X(t) - .0198\sqrt{1.6650 - .0063\ X(t) - .0396\ Y(t)\ X(t) - .3204\ Y(t) - Y(t)^2 - .0004\ X(t)^2} & \tau < .2130 \\ .3477\ Y(t) + .4762\ X(t) - 1.0877 + .4762\sqrt{2.4351\ X(t) + 1.4603\ Y(t)\ X(t) - 3.3351\ Y(t) - Y(t)^2 - .1898 - .5331\ X(t)^2} & \tau < .3254 \\ .0004\ Y(t) - .1601 - .0198\ X(t) - .0198\sqrt{1.6650 - .0063\ X(t) - .0396\ Y(t)\ X(t) - .3204\ Y(t) - Y(t)^2 - .0004\ X(t)^2} & \text{otherwise} \end{cases}$$

> plot(subs(X(t)=Xi,Y(t)=Eta,[xi,eta,t=0..Tf])); Graphical confirmation of correctness of result is not presented for the sake of brevity.

Computing and plotting of trajectory, velocity and acceleration of scrapers

If we know position of the propelled and driven machine element of the supporting bar, we can use just a few simple relations from linear algebra to derive functions describing their positions as a functions of time. If these functions are known, computation of the elements of vectors of velocity and acceleration or their absolute values is a simple problem only. Because the functions *Ξ*(*t*) and *H*(*t*) and *ξ(τ)* and *η*(*t*) are quite long and their derivatives will be even more complicated, we shall use following substitutions to simplify next computation. But Maple outputs will be still too long, so we shall not print them out.

> Xi1:=diff(Xi,t): Xi2:=diff(Xi,t,t): Eta1:=diff(Eta,t): Eta2:=diff(Xi,t,t): computation of the derivatives
> DSu:=[X(t)='Xi', diff(X(t),t)='Xi1', diff(X(t),t,t)='Xi2', Y(t)='Eta', diff(Y(t),t)='Eta1', diff(Y(t),t,t)='Eta2']; substitutution of the derivatives
> xi1:=subs(DSu,diff(xi,t)): xi2:=subs(DSu,diff(xi,t,t)): eta1:=subs(DSu,diff(eta,t)): eta2:=subs(DSu,diff(eta,t,t)): substituted derivatives
> dsu:=[x(t)='xi', diff(x(t),t)='xi1', diff(x(t),t,t)='xi2', y(t)='eta', diff(y(t),t)='eta1',diff(y(t),t,t)='eta2']; substitution of derivated functions.

We have three scrapers, so we use indexed variables. Significance of the variables corresponds to their names. The first index indicates the first scraper seen from the left.

```
> Ex:=(x(t)-X(t))/S; Ey:=(y(t)-Y(t))/S;
> for j from 1 to 3;
  Fx[j]:=X(t)+Ex*Ph[j]+Ey*Lh; Fy[j]:=Y(t)+Ey*Ph[j]-Ex*Lh;
  Vx[j]:=diff(Fx[j],t); Vy[j]:=diff(Fy[j],t);
  Ax[j]:=diff(Vx[j],t); Ay[j]:=diff(Vy[j],t);
  V[j]:=sqrt(Vx[j]^2+Vy[j]^2); A[j]:=sqrt(Ax[j]^2+Ay[j]^2);
  At[j]:=diff(V[j],t); An[j]:=sqrt(A[j]^2-At[j]^2).
```

Plotting of results

For the sake of brevity we shall present a few graphical outputs only. If we shall create plots in the phase space, we shall use blue color for the *x* – axis and black color for the *y* – axis. The most thick line will indicate the first scraper.

> opt:=linestyle=[2,2,2,3,3,3],thickness=[3,2,1,3,2,1],color=black:
> plot(subs(dsu,DSu,[seq([Fx[j],Vx[j],t=0..Tf],j=1..3),seq([Fy[j],Vy[j], t=0..Tf],j=1..3)]),opt); Phase space plot: *[Fx(t), Vx(t)]* and *[y(t), Vy(t)]* – Figure 3.



Source: own elaboration. Źródło: opracowanie własne.

Fig. 3. Phase space plot: [Fx(t), Vx(t)] – blue, [y(t), Vy(t)] – black
Rys. 3. Przestrzenne wykresy fazowe: [Fx(t), Vx(t)] – niebieski, [y(t), Vy(t)] – czarny

> plot(subs(dsu,DSu,[seq([Fx[j],Ax[j],t=0..Tf],j=1..3),seq([Fy[j],Ay[j], t=0..Tf],j=1..3)]), opt); Phase space plot: *[Fx(t), Ax(t)]* and *[Fy(t), Ay(t)]* – Figure 4.



Source: own elaboration. Źródło: opracowanie własne.

Fig. 4. Phase space plot: [Fx(t), Ax(t)] – blue, [Fy(t), Ay(t)] – black
Rys. 4. Przestrzenne wykresy fazowe: [Fx(t), Ax(t)] – niebieski, [Fy(t), Ay(t)] – czarny

> plot(subs(dsu,DSu,[seq([Vx[j],Ax[j],t=0..Tf],j=1..3),seq([Vy[j],Ay[j], t=0..Tf], j=1..3)]), opt); Phase space plot: *[Vx(t), Ax(t)]* and *[Vy(t), Ay(t)]* – Figure 5.



Source: own elaboration. Źródło: opracowanie własne.

Fig. 5. Phase space plot: [Vx(t), Ax(t)] – blue, [Vy(t), Ay(t)] – black
Rys. 5. Przestrzenne wykresy fazowe: [Vx(t), Ax(t)] – niebieski, [Vy(t), Ay(t)] – czarny

> plot(subs(dsu,DSu,[seq(Fx[j],j=1..3),seq(Fy[j],j=1..3)]),t=0..Tf,opt); *Fx(t)* and *Fy(t)*, one period – Figure 6.



Source: own elaboration. Źródło: opracowanie własne.

Fig. 6. Phase space plot: Fx(t) – black, Fy(t) – blue, one period
Rys. 6. Przestrzenne wykresy fazowe: Fx(t) – czarny, Fy(t) – niebieski, jeden cykl

> opt:=linestyle=[2,2,2,3,3,3,1,1,1],thickness=[3,2,1,3,2,1,6,4,3], color=black:
> plot(subs(dsu,DSu,[seq(Vx[j],j=1..3),seq(Vy[j],j=1..3),seq(V[j],j=1..3)]), t=0..Tf,opt); *Vx(t), Vy(t)* and |*V(t)*| – Figure 7.



Source: own elaboration. Źródło: opracowanie własne.

Fig. 7. Phase space plot: Vx(t) – red, Vy(t) – blue, |V(t)| – black
Rys. 7. Przestrzenne wykresy fazowe: Vx(t) – czerwony, Vy(t) – niebieski, |V(t)| – czarny

```
> plot(subs(dsu,DSu,[seq(Ax[j],j=1..3),seq(Ay[j],j=1..3),seq(A[j],j=1..3)]), t=0..Tf,opt);
```
Ax(t), *Ay(t)* and |*A(t)*| – Figure 8.



Source: own elaboration. Źródło: opracowanie własne.

Fig. 8. Phase space plot: Ax(t) – red, Ay(t) – blue, |A(t)| – czarny
Rys. 8. Przestrzenne wykresy fazowe: Ax(t) – czerwony, Ay(t) – niebieski, |A(t)| – czarny

```
> plot(subs(dsu,DSu,[seq(At[j],j=1..3),seq(An[j],j=1..3),seq(A[j],j=1..3)]), t=0..Tf,opt);
```
At(t) = tangent acceleration, *An(t)* = normal acceleration, |*A(t)*| – Figure 9.



Source: own elaboration. Źródło: opracowanie własne.

Fig. 9. Phase space plot: At(t) = tangent acceleration – red, An(t) = normal acceleration – blue, |A(t)| – black
Rys. 9. Przestrzenne wykresy fazowe: At(t) = przyspieszenie styczne – czerwony, An(t) = przyspieszenie normalne – niebieski, |A(t)| – czarny

> plot(subs(dsu,DSu,[seq(Ax[j]*Vx[j]+Ay[j]*Vy[j],j=1..3), sum(Ax[i]*Vx[i]+Ay[i]*Vy[i], i=1..3)]),t=0..Tf,thickness=[3,2,1,5], color=black); If values of *y* – axis will be multiplied by the mass of the scraped dung, we can determine requested power for one scraper and total power – Figure 10.



Source: own elaboration. Źródło: opracowanie własne.

Fig. 10. Determine requested power for one scraper and total power P(t)
Rys. 10. Determinanty zadanej mocy dla jednego pręta oraz mocy całkowitej P(t)

We can see animation of the working machine. The whole animation is combined from three partial plots.

> A1:=display([seq(plot(subs(dsu,DSu,t=tt,{[[Xi,Eta],[xi,eta]], seq([[Fx[j],Fy[j]],[Fx[j]-Ey*Lh,Fy[j]+Ex*Lh]],j=1..3)}),color=brown), tt=[seq(Tf*i/200,i=0..200)])],insequence=true,thickness=3, scaling=constrained): Supporting bar and scrapers
> A2:=display([seq(plot(subs(dsu,DSu,[seq([Fx[j],Fy[j],t=0..Tf*i/200], j=1..3)]), color =[black,blue,red]),i=1..200)],insequence=true,thickness=3, scaling=constrained): Trajectory of the scrapers
> A3:=display([seq(plot(subs(dsu,DSu,{[Xi,Eta,t=0..Tf],[xi,eta,t=0..Tf]}), scaling= constrained,color=grey,thickness=3),i=0..200)],insequence=true): Trajectory of the propelled and driven machine element.
> AT:=display({A1,A2,A3}): AT; Build whole animation and display it.

Analysis of speed and acceleration of mowing fingers end on feed gear

The speed of cradle element is decisive for right transport of stalk material. There is a time behavior of speed *Vx*, *Vz* and of general speed *v* on the Figure 3. The speed factor *Vz* can affect smoothness of feeding but the general speed *v* is not affected. Absolute speed of cradle element is decisive for smoothness of transport and insertion of material. Absolute speed of first and second mowing finger shows two marked peaks. Those peaks are not significant for even material movement, because more than 70% of material is transported by first and second cradle. It is important the maximal loaded force for dimensioning of constituent elements. Extent of the maximal loaded force is influenced by maximal acceleration. Time behavior of acceleration indicated at Figure 4 is even at the first and second cradle, only at the last cradle there is marked difference. The value of acceleration reaches 78 $m \cdot s^{-2}$ there. It has been considered as the maximum value during the calculation. The figure shows an analysis of tangential non-normal acceleration. It assesses the extent of individual constituents of acceleration.

Conclusion

The presented access can be used if the intersection of the circle = *e*1 – describing all possible positions of the driven machine element with the leading = *e*2 can be computed analytically, together with velocities an accelerations. If the analytical solution cannot be found, we can use following approach.

> e1:=(X(t)-x(t))^2+(Y(t)-y(t))^2=Lambda^2: *X(t)* and *Y(t)* are functions describing position of the propelled machine element, *x(t),* and *y(t)* of the driven machine element.
> e2:=F(x(t),y(t))=0; General equation describing leading.
> sol:=solve({e1,e2},{x(t),y(t)}): If *t* has a numeric value, this returns real values of *x(t) = x* and *y(t) = y*, where *x* and *y* will be real numbers,

sol:={x(t)=x,y(t)=y}

> su:=[diff(x(t),t)=Vx,diff(x(t),t,t)=Ax,y(t)=y,diff(y(t),t)=Vy, diff(y(t),t,t)=Ay]; Substitutions of the first and second time derivatives of *x(t)* and *y(t),*

$$su := \left[\frac{\partial}{\partial t} x(t) = Vx, \frac{\partial^2}{\partial t^2} x(t) = Ax, \frac{\partial}{\partial t} y(t) = Vy, \frac{\partial^2}{\partial t^2} y(t) = Ay \right]$$

> DSu:=[X(t)=X,diff(X(t),t)=Xt,diff(X(t),t,t)=Xtt,Y(t)=Y,diff(Y(t),t)=Yt, diff(Y(t),t,t)=Ytt]; Substitutions of the first and second time derivatives of *X(t)* and *Y(t).*

$$DSu := \left[X(t) = X, \frac{\partial}{\partial t} X(t) = Xt, \frac{\partial^2}{\partial t^2} X(t) = Xtt, Y(t) = Y, \frac{\partial}{\partial t} Y(t) = Yt, \frac{\partial^2}{\partial t^2} Y(t) = Ytt \right]$$

> DFs:=[D[1](F)(x(t),y(t))=Fx,D[1,1](F)(x(t),y(t))=Fxx,D[2](F)(x(t),y(t))=Fy, D[2,2](F)(x(t),y(t))=Fyy,D[1,2](F)(x(t),y(t))=Fxy]; Substitutions of the first and second derivatives of the function *F(x(t), y(t))* derivates by the first *x(t)* or second argument *y(t).*

$$DFs:=[D_1(F)(x(t),y(t))= Fx, D_{1,1}(F)(x(t),y(t))= Fxx, D_2(F)(x(t),y(t))= Fy, D_{2,2}(F)(x(t),y(t))= Fyy, D_{1,2}(F)(x(t),y(t))=Fxy]$$

These substations will be used to simplify final terms. We have to remember that substitutions *su*, *DSu* and *DFs* are substitutions of the real numbers. So the final two presented terms *Velocity* and *Acceleration* are methods how to compute velocity and acceleration of the driven machine element.

```
> Velocity:=subs(DSu,DFs,su,sol,solve({diff(e1,t),diff(e2,t)},{diff(x(t),t), diff(y(t),t)}));
```

$$\text{Velocity} := \left\{ Vx = \frac{Fy\left(Xt\left(x-X\right)+Yt\left(y-Y\right)\right)}{Fx\left(Y-y\right)+Fy\left(x-X\right)},\quad Vy = \frac{Fx\left(Xt\left(x-X\right)+Yt\left(y-Y\right)\right)}{Fx\left(Y-y\right)+Fy\left(x-X\right)} \right\}$$

```
> Acceleration:=subs(DSu,DFs,su,sol,solve({diff(e1,t,t),diff(e2,t,t)},
  {diff(x(t),t,t),diff(y(t),t,t)}));
```

$$\text{Acceleration} := \left\{ Ax = \frac{\left(Fy\,Ytt + Fyy\,Vy^2 + Fxx\,Vx^2 + 2\,Vx\,Fxy\,Vy\right)\left(Y-y\right) + \left(\left(Xt-Vx\right)^2 + \left(Yt-Vy\right)^2 + Xtt\left(X-x\right)\right)Fy}{\left(y-Y\right)Fx + \left(X-x\right)Fy},\right.$$
$$\left. Ay = \frac{\left(Fx\,Xtt + 2\,Vx\,Fxy\,Vy + Fxx\,Vx^2 + Fyy\,Vy^2\right)\left(x-X\right) - \left(\left(Xt-Vx\right)^2 + \left(Yt-Vy\right)^2 + Ytt\left(Y-y\right)\right)Fx}{\left(y-Y\right)Fx + \left(X-x\right)Fy} \right\}$$

Advantageous mathematical solution of kinematics by Maple program follows from initiated solution of feed gear of pick-up cylinder. This method can be used in construction design of functional groups, in solidity calculation of individual elements and in their dimensioning in agricultural operation. The calculations can help to reveal critical points of movement of matter into press mechanism for already manufactured machines. The right functions are possible to verify by computer animation of initiated kinematics mechanism.

References

CAMPION G., BASTIN B., D'ANDREA-NOVEL B. 1996. Structural properties and classification of kinematic and dynamic models of wheeled mobile robots. IEEE Transactions on Robotics and Automation. Vol. 12. No. 1 p. 47–62.

CALKINS D. E., EGGING N., SCHOLZ C. 2000. Knowledge-based engineering (KBE) design methodology at the undergraduate and graduate levels. International Journal of Engineering Education. Vol. 16. No. 1 p. 21–38.

DINDORF R., MŁYNARSKI T., WOŁKOW J. 2000. Modelowanie kinematyki i dynamiki hydraulicznego mechanizmu podnoszenia [Kinematic and dynamic modeling of hydraulic lifting mechanism]. Maszyny Dźwigowo-Transportowe. Nr 1 p. 22–30.

GECOW M. 2012. Modelowanie i animacja układów napędowych parowozów w programie 3DS Max [Modeling and animation systems of drive locomotives in the 3DS Max]. Praca dyplomowa inżynierska. Maszynopis. Warszawa. Politechnika Warszawska. Wydział Elektroniki i Technik Informacyjnych. Instytut Informatyki pp. 66.

HAŁAT W. 2007. Zastosowanie komputerowego rachunku symbolicznego do zagadnień zginania belek [The use of the computers symbolik accont in the field of bending beams]. Górnictwo i Geoinżynieria. Z. 3 p. 171–182.

IGNATEV A. M., KOLESOV A. V., LEVCHENKO P. S., PODGORNOV A. V. 2015. Исследование кинематики движения механизма с помощью параметрических программных продуктов [Study of the kinematics mechanism motion using parametric programe]. Материалы международной научно-технической конференции ААИ «Автомобиле- и тракторостроение в России: приоритеты развития и подготовка кадров», посвященной 145-летию МГТУ «МАМИ».Секция 6 «Машины и технологии заготовительного производства», Подсекция «Машины и технологии обработки металлов давлением» [online]. [Access 23.12.2015]. Available at: http://www.mami.ru/science/mami145/scientific/article/s06/s06_14.pdf

JAKUBIAK J., MUSZYŃSKI R. (tech. ed.). 2012. Narzędzia komputerowe w robotyce. Modelowanie kinematyki i dynamiki [Computer tools in robotics. Modeling of kinematics and dynamics] [online]. Projekt przejściowy 2011/2012 specjalności Robotyka na Wydziale Elektroniki. Wrocław. Politechnika Wrocławska. Zakład Podstaw Cybernetyki i Robotyki. pp. 230. [Access 23.12.2015]. Available at: http://sequoia.ict.pwr.wroc.pl/~mucha/TechKomp/narzedzia_komputerowe_w_robotyce_proj_przejsciowy_2011.pdf

MALYGIN A. 2012. Проектирование и анимация кинематических схем механизмов в САПР CATIA [Design and kinematic simulation of mechanical systems in SAP CATIA] [online]. [Access 23.12.2015]. Available at: http://cadregion.ru/catia/proektirovanie-i-animaciya-kinematicheskix-sxem-mexanizmov-v-sapr-catia.html

MITUŚ A.C., ORLIK R., PAWLIK G. 2007. Wstęp do pakietu algebry komputerowej Maple [Introduction to computer Maple algebra package] [online]. [Access 23.12.2015]. Available at: http://www.if.pwr.edu.pl/~grpawlik/listyz/MapleACM.pdf ss. 70.

OMBACH J. 2006. Wprowadzenie do metod probabilistycznych wspomagane komputerowo – Maple [Introduction to probabilistic methods supported by computer – Maple]. Nowy Sącz. PWSZ. ISBN 83-88887-81-5 pp. 171.

PAZDERSKI D. 2012. Modelowanie kinematyki robotów kołowych. Kurs algorytmy sterowania ruchem robotów mobilnych [Modeling of kinematics wheeled robots. Course of algorithms controling motion of mobile robots] [online]. [Access 23.12.2015]. Available at: http://etacar.put.poznan.pl/dariusz.pazderski/data/RM_ster2.pdf

PAZDERSKI D., KOZŁOWSKI K. 2008. Trajectory tracking of undeactuated skid-steering robot. Proceedings of American Control Conference. Seattle, USA, June 11–13 2008 p. 3506–3511.

SAPRONOV I. 2015. Анимация и оптимизация при анализе кинематических механизмов в T-FLEX CAD [Simulation and optimization in the process of analysis of kinematic mechanisms by T-FLEX CAD] [online]. [Access 23.12.2015]. Available at: https://www.youtube.com/watch?v=hAz46g1dPy4

YOUNG H.D., FREEDMAN R.A. 1996. University physics. New York. Addison-Wesley. ISBN 0-201-84769-8 pp. 1400.

ZAWADZKI P., GÓRSKI F., KOWALSKI M., PASZKIEWICZ R., HAMROL A. 2011. System for 3D models and technology process design. Transactions of FAMENA. Vol. 35. Iss. 2 p. 69–78.

ZAWADZKI P., KOWALASKI M., WICHNIAREK R., KLIŃSKI G. 2013. Automatyzacja procesu projektowania rur giętych w oparciu o parametryczny system CAD [Automation of the design process of flexible pipes on the basis of parametric CAD system] [online]. Artykuł autorski. Wydarzenie Targowe. CAxInnovation, Targi Innowacje Technologie Maszyny. Poznań, 4–7 czerwca 2013 r. [Access 23.12.2015]. Available at: http://www.docplayer.pl/3823267-Automatyzacja-procesu-projektowania-rur-gietych-w-oparciu-o-parametryczny-system-cad.html

Acknowledgement

Authors would like to thank Autocont Brno and Waterloo Maple for their continuous interest and support.
Computed on computer Autocont, PIV, 512 MB RAM
Maple jest zastrzeżonym znakiem towarowym University of Waterloo.

Jan Červinka, Stanislav Bartoň, Edmund Kamiński, Bronisław Puczel

MODELOWANIE KINEMATYKI MECHANIZMÓW ZA POMOCĄ PROGRAMU KOMPUTEROWEGO MAPLE

Streszczenie

W artykule przedstawiono, jak wykorzystywać program Maple 7 w celu analizowania kinematyki maszyny złożonej. Rozpatrywano trajektorie, prędkości i przyśpieszenia elementów roboczych. Wykonano obliczenia komputerowe, a wyniki przedstawiono graficznie, pokazując kolejne fazy. Zalety metody matematycznego rozwiązywania zadania kinematycznego za pomocą programu Maple przedstawiono na przykładzie rozwiązania bębna podbieracza. Metoda ta może być wykorzystana do funkcjonalnego łączenia zestawów maszyn lub obliczeń wytrzymałościowych poszczególnych elementów i ustalania ich gabarytów. Obliczenia mogą być pomocne do określania krytycznych punktów wewnątrz mechanizmów prasy zbierającej z możliwością wprowadzenia zmian do maszyn już produkowanych. Komputerowa animacja kinematyki mechanizmu umożliwia sprawdzenie poprawności jego funkcjonowania.

Słowa kluczowe: technika analogowa, program Maple, podbieracz bębnowy, kinematyka

Adres do korespondencji:
prof. dr hab. Edmund Kamiński
Instytut Technologiczno-Przyrodniczy
Mazowiecki Ośrodek Badawczy w Kłudzienku
05-825 Grodzisk Mazowiecki
tel. 22 755-60-41 wew. 112; e-mail: e.kaminski@itp.edu.pl